
Planets/Planet.pde
class Planet{
  float diameter;
  color surface;
  float angle;

  Planet(float diameter, color  surface){
    this.angle = random(350);
    this.diameter = diameter;
    this.surface = surface;
  }

  public void orbit(float radiusX, float radiusY, float orbitTime){
    // draw ellipse first, so it's under the planet
    drawOrbit(radiusX, radiusY);
    float px = cos(radians(angle))*radiusX/2;
    float py = sin(radians(angle))*radiusY/2;
    
    // draw planet next 
    fill(surface);
    noStroke();
    ellipse(px, py, diameter/2000, diameter/2000);
    
    //big planets get a red dot award
    if (diameter > 100000){
        fill(#FF5522); 
        ellipse(px, py+20, 10, 10);
    }
    
    //move forward
    angle += (1/orbitTime);
  }

  void drawOrbit(float radiusX, float radiusY){
       //make the entire orbit visible
       stroke(255, 50);
       float angle=0;
             for (int i=0; i<360; i++){
                  point(cos(radians(angle))*radiusX/2, 
                        sin(radians(angle++))*radiusY/2);
             }
       }
}




Planets/Planets.pde
//source: http://processing.org/discourse/beta/num_1162399456.html
//posted by ira, (in reply to: Building a graphic of a solar system)
//adapted by (c) 2012 Loe Feijs TU/e

Planet mercury;
Planet venus;
Planet earth;
Planet mars;
Planet jupiter;

void setup(){
  size(800, 600);//,P3D);
  mercury = new Planet(4878, #DDDDDD); 
  venus = new Planet(12104, #6688FF); 
  earth = new Planet(12756, #00FF00); 
  mars = new Planet(6788, #FF0000); 
  jupiter = new Planet(142796, #FFEE33); 
}

void draw(){
  background(0);
  translate(width/2, height/2);
  //rotateX(radians(55));
  mercury.orbit(58, 59, 0.24); //min and max distance to sun in 1000000km
  venus.orbit(108, 109, 0.62); //rotation time in years
  earth.orbit(150, 153, 1.00); 
  mars.orbit(230, 232, 1.88);
  jupiter.orbit(778, 782, 11.87);
}





